

6.946 Driven Pendulum Project

Dennis V. Perepelitsa

18 November 2006

Gerald Jay Sussman, Jack Wisdom

Exercise 3.14: Periodically driven pendulum

We investigate several regions of parameter space of a driven pendulum system over which specific phenomena occur. The pivot on which the pendulum hangs is driven in the vertical direction by a periodic force $A \cos(\omega t)$. We reuse code relevant to the driven pendulum system in SICM: Section 1.6.2 contains the Lagrangian for a driven pendulum in general, Section 1.7 provides for the specific case of the periodically driven pendulum, and Section 3.5 derives the Hamiltonian.

The pendulum we investigate has $m = 1$ kg, $g = 9.8$ m/s² and $l = 9.8$ m. The period (ω) and amplitude (A) of the drive are variable.

Determination of the Parameter Space

We are asked to find the subset of a two-dimensional parameter space over which a certain predicate is true. To do this, we develop a series of tools and use them in conjunction with some simplifying assumptions. The following procedure returns a region of parameter space between some range of A and ω , with some step-size dA and $d\omega$.

```
(define (get-square-list xstart xend xdelta ystart yend ydelta)
  (if (>= xstart xend)
      (get-line-set-x ystart yend ydelta xstart)
      (append (get-line-set-x ystart yend ydelta xstart)
              (get-square-list (+ xstart xdelta) xend xdelta ystart yend ydelta))))

(define (get-line-set-x ystart yend ydelta x)
  (glsx-helper ystart yend ydelta x)
  )

(define (glsx-helper cury endy stepy x)
  (if (> (+ cury stepy) endy)
      (list (list x cury))
      (cons (list x cury) (glsx-helper (+ cury stepy) endy stepy x))))
```

Now we write a plotting procedure that plots a region of parameter space.

```
(define (plot-x-y-pairs win x)
  (if (null? x)
      #t
      (let ()
        (plot-point win (caar x) (cadar x))
        (plot-x-y-pairs win (cdr x)))))
```

Modification to System Evolution

To examine the trajectories caused by the surface of section method, we want to save them to a list for later use. All of our stroboscopic trajectories are obtained using code like the following.

```
(define (save-point point)
  (let ((t (time point))
        (theta (coordinate point))
        (p (momentum point)))
    (set! collected-points
          (append collected-points
                  (list (list t theta p))))))

(set! collected-points (list))

((evolve H-pend-sysder 1 9.8 9.8 some-value-of-A some-value-of-omega)
 (up 0 0 p) save-point (/ :2pi w) (* 50 :2pi) 1.0e-12)
```

(a) Determination of the Inverted Equilibrium Parameter Space

The pendulum stays upright around $\theta = \pi$, its tendency to fall down countered by the drive. An example of this phenomenon is pictured in Figure 1. Now we need a (mostly) deterministic way to test whether a given set of parameters (A, ω) exhibit this behavior. This is a harder problem than it seems but the surface of section method makes the presence of the phenomena somewhat apparent.

The inverted equilibrium phenomenon is characterized by surface of section trajectories orbiting around the inverted $\theta = \pm\pi$ point with small momentum. Thus, if an inverted equilibrium exists, trajectories with initial points close to $(\pi, 0)$ will stay within that region of the phase plane. We use the following process to test for the predicate:¹

1. Set the initial point is at $\theta_0 = \pi$, with a very small initial positive p_0 on the order of .1 or .01.
2. Compute the surface of section for a large number of points (on the order of a hundred).
3. If any of them have significantly higher ($\sim 1.1p_0$, to allow for rounding error) absolute momentum than that of the initial point, this indicates that the trajectory has left the area, and the predicate is false for this parameter set.

In Scheme, we implement this test as a procedure.

```
(define (list-escapes-p p l)
  (fold-right (lambda (x y) (or x y)) #f
             (map (lambda (x) (< (abs p) (caddr x))) l)))
```

¹An annoying detail about this approach is that false positives are possible. It is possible (although very unlikely), that the next several hundred points in a chaotic section will happen to fall along the small region $-1.1 p_0 < p < 1.1 p_0$, but we ignore this.




Figure 1: Two example of the inverted equilibrium phenomenon. The plot on the left is a surface of section for $A = 3$, $\omega = 6$. The extremes of the ordinate axis are -500 and 500 . The plot on the right is a surface of section for $A = 4$, $\omega = 7$, with the ordinate ranging from -700 to 700 .

```
(define (does-this-escape A w theta p ptest)
  (let ()
    (set! collected-points (list))
    ((evolve H-pend-sysder 1 9.8 9.8 A w)
     (up 0 theta p) save-point (/ :2pi w) :2pi 1.0e-12)
    (list-escapes-p ptest collected-points)))
```

We coarsely sample a relatively broad section of the parameter space, and see that the most interesting region is when ω and A are less than ten or so. We choose a relatively small granularity to get a good definition, and decide to be generous on our definition of “escapes”, setting the initial point to $p_0 = .01$, and the threshold to $10p_0$.

```
(define my-plot (frame 0 10 0 10))

(plot-x-y-pairs my-plot
  (filter (lambda (l) (not (does-this-escape (car l) (cadr l) :-pi .01 .1)))
    (get-square-list 1 10 .05 1 10 .05)))
```

The results are pictured in Figure 2. It appears the region of parameter space extends into higher values of A as well.



Figure 2: Region of parameter space for which an inverted equilibrium exists. The ordinate is $0 \leq \omega \leq 10$. The abscissa is $0 \leq A \leq 10$. The points are spaced $\Delta A = \Delta \omega = .05$ apart.

(b) Investigation of Period Doubling

The pendulum tips to one side and then the other as the support is driven, visiting either side with a period equal to twice the natural period. An example of this phenomenon is pictured in Figure 3. We are interested in finding the angular momenta at which these equilibria lie at different values of the driving frequency for a particular value of the driving amplitude $A = .4$.

First we write procedures to determine the angular momentum associated with a given amplitude and driving frequency. Specifically, assuming we have hit upon an elliptic period-doubled orbit, we can make a good guess at the momentum by averaging the highest and lowest momenta reached by the top half of the trajectory. The following code accomplishes this.

```
(define (get-center A w p)
  (let ()
    (set! collected-points (list))
    ((evolve H-pend-sysder 1 9.8 9.8 A w)
     (up 0 0 p) save-point (/ :2pi w) (* 50 :2pi) 1.0e-12)
    (extract-center-helper collected-points)))

(define (extract-center-helper cp)
  (let ((pts (filter (lambda (x) (> x 0)) (map caddr cp))))
    (/ (+ (fold-right max 0 pts) (fold-right min 10000 pts)) 2.0)))
```

Now, by taking pairs of the amplitude A and a guess at the associated momentum $p(A)$ and returning the amplitudes with an updated guess at the momentum, we can iteratively improve our guess of the momenta by feeding the output back into the procedure below.




Figure 3: A classic example of period doubling in a surface of section, with parameters $A = .4$, $\omega = 2$. The axes are $-300 \leq p \leq 300$ and $-\pi \leq \theta \leq \pi$. Both stable and unstable equilibria exhibit the phenomenon in this example.

```
(define (improve-iteration pts)
  (if (= (length pts) 1)
      (list (list (caar pts) (get-center (caar pts) 2 (cadar pts))))
      (cons (list (caar pts) (get-center (caar pts) 2 (cadar pts)))
            (improve-iteration (cdr pts)))))
```

Investigating individual points by hand, we find that the range $1.3 < \omega < 2.1$ is likely to exhibit the phenomenon, choose a relatively small step-size of $\Delta\omega = .001$, and compute several iteratively improved plots of $p(\omega)$. The code below accomplishes this.

```
(define (get-range a b delta)
  (if (> (+ a delta) b)
      (list a)
      (cons a (get-range (+ a delta) b delta))))

(define my-plot (frame 1.3 2.1 0 200))

(plot-x-y-pairs my-plot
  (improve-iteration
   (map (lambda (x) (list x (- 200 (* 70 x))))
        (get-range .1 2.1 .005)))
  ))
```




Figure 4: Plots of the momentum of the stable equilibrium $p(\omega)$ for $A = .4$. On both plots, the ordinate ranges from $0 \leq p \leq 200$, and the abscissa from $1.3 \leq \omega \leq 2.1$. The plot on the left shows the first four (converging) iterations. The plot on the right shows only the last one.

The results are shown in Figure 4. The graphical evidence is that as ω increases, the period-doubled region emerges suddenly, and then slowly integrates back into the center.

We repeat the procedure, changing the initial condition in the definition of `get-center` to investigate elliptic trajectories around the $\theta = \pi$ unstable point. Because of strange behavior at the ends of the range of ω for which the unstable period doubled region exists, we'll examine only the behavior in the middle. Figure 5 presents the quickly-converging results. Graphically, the momentum of the unstable equilibrium seems to rise linearly with driving frequency in this range.

(c) Determination of the Large-scale Chaos Parameter Space

This phenomenon is characterized by the presence of three large, distinct chaotic regions. A trajectory starting inside one of the three never seems to leave and enter the other two. An example of a pair of coordinates that exhibit this is shown in Figure 6.

Since it is rarer, we will investigate the lack of linkage between large-scale chaos zones, rather than its presence.

As before, we want to derive a test, that is, a procedure, that determines whether a given parameter tuple (A, ω) exhibits this phenomenon. The only useful way we can really check for transition to large-scale chaos is to let the system run for a large number of iterations, and see if the range of momentum values that the surface of section takes exceed some range.

Trying some values of (A, ω) by hand, we see that the separated chaotic-zones phenomenon tends to occur at low values of A and high values of ω . We see by experiment that the central chaotic zone does not typically range higher than $p \approx 250$ or, more conservatively, $p = 300$. Thus, any orbits starting just off the separatrix (say at $\theta = \pi$, $p = .01$), will be only be found above




Figure 5: Plot of the momentum of the unstable equilibrium $p(\omega)$ for $A = .4$. The ordinate ranges from $0 \leq p \leq 200$, and the abscissa from $1.4 \leq \omega \leq 2.5$. The plot shows two (converging) iterations. More iterations lead to the same linear relationship.




Figure 6: Examples of principal chaotic zone separation. The left plot has $\omega = 8$ for $A = 1$, the ordinate ranges from $-600 \leq p \leq 600$, and the abscissa from $-\pi \leq \theta \leq \pi$. The plot on the right has $A = .5$, $\omega = 7$, and the axes are $-\pi \leq \theta \leq \pi$ and $-900 \leq p \leq 900$.

$p = 300$ if they have entered the “top” chaotic zone. We use this as our (unfortunately somewhat imprecise) test, and write a procedure that returns the maximum and minimum momenta reached after a few hundred iterations starting just off the unstable equilibrium point.

```
(define (max-min-p-from-separatrix A w)
  (let ()
    (set! collected-points (list))
    ((evolve H-pend-sysder 1 9.8 9.8 A w)
     (up 0 (+ :-pi .01) 0) save-point (/ :2pi w) (* 250 :2pi) 1.0e-12)
    (extract-max-and-min-p collected-points)))

(define (extract-max-and-min-p cp)
  (let ((pts (map caddr cp)))
    (list (fold-right max 0 pts) (fold-right min 10000 pts))))
```

Now all that remains is to set up our plot, choose a region of parameter space (we choose $.1 \leq A \leq 2$, $4 \leq \omega \leq 10$, with spacing $\Delta A = .05$ and $\Delta \omega = .2$), and run the procedures.

```
(define my-plot (frame .1 2 4 10))

(map
 (lambda (x) (plot-point my-plot (caar x) (caddr x)))
 (filter
  (lambda (x) (> (caadr x) 300))
  (map
   (lambda (x) (list x (max-min-p-from-separatrix (car x) (cadr x))))
   (get-square-list .1 2 .05 4 10 .2))))
```

The results are shown in Figure 7. Ostensibly, this is the region of the parameter space for which the three chaotic regions around the principal islands are separate. However, this is a crude approximation at best - I could not think of a better indication of chaotic zone separation than this first order one.

Though it is computationally expensive to get definition better $\Delta A = .05$, $\Delta \omega = .2$, the region of parameter space is still fairly apparent.




Figure 7: Region or parameter space for which the principal chaotic regions are not linked. The ordinate is $.1 \leq A \leq 2$ and the abscissa is $4 \leq \omega \leq 10$.